<commit_before>{ stdenv, fetchzip }:

let
  version = "0.98";

in fetchzip {
  name = "onestepback-${version}";

  url = "http://www.vide.memoire.free.fr/perso/OneStepBack/OneStepBack-v${version}.zip";

  postFetch = ''
    mkdir -p $out/share/themes
    unzip $downloadedFile -x OneStepBack/LICENSE -d $out/share/themes
  '';

  sha256 = "0sjacvx7020lzc89r5310w83wclw96gzzczy3mss54ldkgmnd0mr";

  meta = with stdenv.lib; {
    description = "Gtk theme inspired by the NextStep look";
    homepage = https://www.opendesktop.org/p/1013663/;
    license = licenses.gpl3;
    platforms = platforms.all;
    maintainers = [ maintainers.romildo ];
  };
}<commit_msg>onestepback: 0.98 -> 0.991<commit_after>{ stdenv, fetchurl, unzip }:

stdenv.mkDerivation rec {
  name = "onestepback-${version}";
  version = "0.991";

  srcs = [
    (fetchurl {
      url = "http://www.vide.memoire.free.fr/perso/OneStepBack/OneStepBack-v${version}.zip";
      sha256 = "1jfgcgzbb6ra9qs3zcp6ij0hfldzg3m0yjw6l6vf4kq1mdby1ghm";
    })
    (fetchurl {
      url = "http://www.vide.memoire.free.fr/perso/OneStepBack/OneStepBack-grey-brown-green-blue-v${version}.zip";
      sha256 = "0i006h1asbpfdzajws0dvk9acplvcympzgxq5v3n8hmizd6yyh77";
    })
    (fetchurl {
      url = "http://www.vide.memoire.free.fr/perso/OneStepBack/OneStepBack-green-brown-v${version}.zip";
      sha256 = "16p002lak6425gcskny4hzws8x9dgsm6j3a1r08y11rsz7d2hnmy";
    })
  ];

  nativeBuildInputs = [ unzip ];

  sourceRoot = ".";

  installPhase = ''
    mkdir -p  $out/share/themes
    cp -a OneStepBack* $out/share/themes/
    rm $out/share/themes/*/{LICENSE,README*}
  '';

  meta = with stdenv.lib; {
    description = "Gtk theme inspired by the NextStep look";
    homepage = http://www.vide.memoire.free.fr/perso/OneStepBack;
    license = licenses.gpl3;
    platforms = platforms.all;
    maintainers = [ maintainers.romildo ];
  };
}
{'fraction_non_alphanumeric': 0.14806980433633, 'fraction_numerical': 0.053410893707033315, 'mean_word_length': 3.3897911832946637, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 4, 'https://': 1, 'lorem ipsum': 0, 'www.': 6, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 1, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '6802451', 'n_tokens_mistral': 825, 'n_tokens_neox': 764, 'n_words': 103}